

José de Francisco Estaire

IDEAS Y RECURSOS

Se presenta un módulo para *Derive* que permite ir más allá del simple dibujo de la gráfica de una función. Tan sólo basta cargarlo y definir en él la función $F(x)$ sobre la que se quiere aplicar, para que las funciones que lo integran proporcionen sus elementos característicos: asíntotas, máximos y mínimos, inflexiones,...

EL ORIGEN de este artículo está en no haber preparado debidamente una clase.

La historia es la siguiente. Había terminado, en 2.º de Ciencias Sociales, el tema «Trazado de Funciones» y realizado algún ejercicio de aplicación, así que decidí hacer un ejercicio completo en el que se trataran todos los apartados estudiados. Del libro de texto, en la sección de ejercicios propuestos seleccioné:

Representar la función

$$y = \frac{x^3 + x^2 - 2}{x^2 - 2x - 4}$$

Calculamos las asíntotas, la derivada primera y después de simplificarla obtuvimos:


$$y' = \frac{x^4 - 6x^3 - 15x^2 - 4x - 6}{(x^2 - 3x - 4)^2}$$

El paso siguiente fue solucionar $y' = 0$. Al tratar de anular el numerador dije: «Ahora buscaremos por Ruffini las raíces del numerador y cruzaremos los dedos para que sean divisores del término independiente». Comenzamos a probar los divisores de 6 sin éxito. En ese momento sonó el timbre que indicaba el final de la clase y me despedí diciendo: «No hay que preocuparse, pues si las raíces son irracionales mi amigo *Derive* nos las encontrará».

Cuando fui a casa abrí el programa y le suministré la función. Calculé la derivada primera, por si hubiera habido algún error en la simplificación, comprobé que no había ocurrido y mande que resolviera la ecuación $y' = 0$. Como sospechaba las raíces eran irracionales; obteniendo:

$$[x = 7.95956, x = -1.85656, x = -0.0514983 + 0.63512 i, x = -0.0514983 - 0.63512 i]$$

Al tener el programa abierto represente la función, pero eche en falta los elementos característicos como asíntotas,



máximos, mínimos, inflexiones... estas ausencias fueron las que me llevaron al deseo de tener un programa tal que, conocida la función permitiese el cálculo de los elementos necesarios para el estudio e interpretación de dicha función, de forma que el trabajo de cálculo solo se hiciera una vez. Para definir las funciones que calculan los apartados deseados tuve que utilizar algunas de las funciones predefinidas en *Derive*, las funciones utilizadas han sido:

SOLVE ; LIM ; IF ; VECTOR ; INSERT_ELEMENT

Recordemos su sintaxis:

SOLVE([F(x, y) = a, G(x,y) =b] ,[x, y]): resuelve el sistema $F(x, y) = a$, $G(x, y) = b$, en las variables x e y .

SOLVE(F(x), x): resuelve la ecuación $F(x)=0$, en la variable x .

LIM(F(x), x, a): calcula el límite de la función $F(x)$ cuando la variable x tiende hacia a . El valor a puede ser mas o menos infinito.

Si queremos calcular los límites laterales cuando x tiende hacia a la sintaxis es:

$LIM(F(x), x, a, 1)$ y $LIM(F(x), x, a, -1)$, que calculan el límite de $F(x)$ cuando x tiende hacia a , siendo $x > a$ (derecha) y siendo $x < a$ (izquierda).

IF(«Condición C», realizar A si se cumple la condición C, realizar B si no se cumple la condición C). Abreviadamente IF(C, A, B).

La función VECTOR puede generar un vector o un vector cuyas componentes sean vectores (una matriz).

VECTOR(Función o funciones en la variable n , variable n , Valores que toma n)

VECTOR ([x, F(x)], x, [1, 2, 5, 9]) genera la matriz cuyas filas son: [1 ,F(1)] , [2 ,F(2)] , [5 ,F(5)] , [9 ,F(9)]

Si queremos que la primera fila de una matriz A no sea numérica y que tenga por valores el encabezamiento de una tabla, debemos insertar una primera fila con dichos encabezamientos mediante INSERT_ELEMENT.

INSERT_ELEMENT([fila a insertar], A, K), siendo A el vector o matriz en la que vamos a insertar la fila y K la posición en la que se realizara la inserción. Como en la primera fila queremos que figure un texto y no un cálculo debemos colocar entre comillas cada elemento de la fila insertada.

Una forma de realizar el estudio es mediante los siguientes pasos:

1.º Definir la función a estudiar

$F(x)$:= «función objetivo»

2.º Cálculo de las asíntotas verticales

VERTICALES(x) := SOLVE(F(x) = ∞, x)

Resuelve la ecuación para la variable x . En algunas funciones (principalmente en las definidas a trozos) no efectúa el cálculo correctamente.

3.º Asíntotas horizontales y oblicuas

ASINTOTAS(x) := IF (LIM(F(x), x, ∞) = ∞, IF(LIM(F(x)/x = ∞, x, ∞), «No tiene Asíntotas», «La Asíntota oblicua es» y = LIM(F(x)/x, x, ∞) * x + LIM(F(x) - LIM(F(x)/x, x, ∞) * x, x, ∞)) , «La Asíntota horizontal es» y = LIM(F(x), x, ∞))

En nuestro caso si el límite de $F(x)$ es infinito calcula $A = \text{límite de } F(x)/x$. Si no es infinito calcula $B = \text{« a Asíntota horizontal es» } y = \text{Valor del Límite}$.

Cuando se cumple la condición nos conduce a otro IF en el que la condición es: si el límite de $F(x)/x$ es infinito, entonces $A = \text{«no tiene Asíntotas»}$, si el cociente no es infinito se cumplirá $B = \text{«la Asíntota oblicua es» } y = mx + b$

4.º La función, su derivada primera y su segunda derivada

Las agrupamos en una matriz de 3×1 . Cada fila de la matriz debe ir entre corchetes y el texto que queremos que aparezca tiene que ir entrecomillado.

$$\text{FUNCIONES}(x) := [[\text{«F(x)»}=F(x)], [\text{«F'(x)»} = F'(x)], [\text{«F''(x)»}=F''(x)]]$$

Esta matriz permite trabajar con cada fila independientemente.

5.º Puntos de corte. Posibles máximos, mínimos y puntos de inflexión

Resolvemos las ecuaciones $F(x) = 0$, $F'(x) = 0$, $F''(x) = 0$. Para ello utilizaremos la función SOLVE entre corchetes ya que de esta forma obtenemos las soluciones como componentes de un vector.

$$\text{CORTES}(x) := [\text{SOLVE}(F(x) = 0, x)]$$

$$\text{MAXIMOS}(x) := [\text{SOLVE}(F'(x) = 0, x)]$$

$$\text{INFLEXIONES}(x) := [\text{SOLVE}(F''(x) = 0, x)]$$

6.º Límites laterales

Si queremos calcular los límites laterales en algunos puntos p_1, p_2, p_3, \dots que consideremos interesantes (puntos de discontinuidad, asíntotas verticales, ...). Les agruparemos en un vector p .

$$\text{LATERALES}(p) := [\text{LIM}(F(x), x, p, -1), \text{LIM}(F(x), x, p, 1)]$$

$$\text{VECTOR}(\text{LATERALES}(p), p, [p], p_2, p_3, \dots) =$$

La última instrucción calcula los límites laterales en los puntos p_1, p_2, p_3, \dots mediante la función LATERALES(p). La variable de la función es p y no x . El resultado es una matriz de dos columnas e i filas, siendo la primera columna el límite por la izquierda en cada punto p_i y la segunda columna es el límite por la derecha.

```

Estudio de Funciones
#18: PUNTOS(x) := [-2, -1, -0.5, 0, 0.346014, 0.5, 0.648865, 1, 2,
2.89005, 3]
#19: TABLA_MINIMA(x) := INSERT_ELEMENT(["x", "F(x)", "Signo de F'(x)",
"F'(x)"], VECTOR([x, F(x), SIGN(F'(x)), F'(x)], x, PUNTOS(x)),1)
#20: TABLA_MINIMA(x) :=
#21: TABLITA(x) := VECTOR([x, F(x)], x, PUNTOS(x))
#22: TABLITA(x) :=

```

"x"	"F(x)"	"Signo de F'(x)"	"F'(x)"
-2	-0.527194	1	0.849368
-1	?	?	?
-0.5	-1.89683	1	12.224
0	0	1	1
0.346014	0.157642	±1	0
0.5	0.131798	-1	-0.322174
0.648864	0.0689785	-1	-0.47852
1	?	?	?
2	7.58733	-1	-4.6367
2.89004	6.34347	±1	0
3	6.35099	1	0.132312

Página: 2

7.º Puntos de estudio y valoración (máximos, inflexiones, ...)

Agrupamos todos los puntos en una función.

$$\text{PUNTOS}(x) := [\text{puntos de estudio y valoración separados por comas}]$$

8.º Tabla de valores de la función y su derivada

En cada punto de estudio podemos calcular $F(x)$, el signo de la primera derivada y el valor de $F'(x)$. Una forma de conseguirlo es mediante una matriz que llamaremos TABLA_MINIMA. Esta matriz tendrá en la primera fila, como encabezamiento: x , $F(x)$, signo de $F'(x)$ y $F'(x)$ y en las filas siguientes las tabulaciones de las funciones en los puntos deseados.

$$\text{TABLA_MINIMA}(x) := \text{INSERT_ELEMENT}([\text{«x»}, \text{«F(x)»}, \text{«Signo de F'(x)»}, \text{«F'(x)»}], \text{VECTOR}([x, F(x), \text{SIGN}(F'(x)), F'(x)], x, \text{PUNTOS}(x)), 1)$$

TABLA_MINIMA, mediante el valor de $F'(x)$ y su signo a la izquierda y derecha de un punto en el que se anula la derivada, permite dilucidar si en ese punto existe máximo o mínimo. El estudio del signo a la izquierda y derecha de un punto es fundamental, pues a veces al estudiar el signo en las soluciones de $F'(x) = 0$ no da signo cero, pues da un valor pequeño a la derivada en el punto y por tanto tiene signo positivo o negativo.

9.º Tabla de valores de la función

Si queremos representar un punto no es posible con TABLA_MINIMA(x), pues tiene cuatro columnas. Necesitaremos una tabla en la que únicamente figuren la variable x y el valor de $F(x)$. Esto lo conseguimos mediante la función TABLITA(x)

TABLITA (x) := VECTOR([x, F(x)], x, PUNTOS(x))

En nuestro caso la función es un vector de dos componentes $[x, F(x)]$, la variable es x y los valores que tomará serán los puntos que hemos seleccionado para el estudio y que figuran en PUNTOS(x).

10.º

Guardamos en la carpeta de *Math* de *Derive* el archivo «Estudio de Funciones» (siendo su extensión *.mth*). Cada vez que queramos estudiar una función llamaremos al archivo, indicamos quien es $F(x)$, copiamos las funciones de estudio y posteriormente bastara con llamarlas para obtener los resultados.

Al aplicar el algoritmo anterior a diversas funciones he notado que en las definidas a trozos al resolver la ecuación:

derivada primera = 0, a veces no se obtienen todas las soluciones, la forma mas eficaz es resolver la primera derivada de cada trozo y seleccionar los puntos que están en el intervalo de definición. En estas funciones el cálculo de las asíntotas verticales suele dar algún problema.

Apliquemos todo lo anterior al estudio de la función (ver figuras):

$$y = xe^{2x/(x^2-1)}$$

Bibliografía

- GARCÍA, A. (1994): *Prácticas de matemáticas con Derive*, Alfonsa García, Madrid.
- GUZMAN, M y B. RUBIO (1993): *Análisis Matemático -3- Sucesiones y series de funciones, números complejos*, Derive, aplicaciones, Pirámide, Madrid.
- IBAÑEZ JALÓN, M, M. F. PÉREZ MARTÍNEZ, A. J. POBLACIÓN SAEZ y A. SUÁREZ BARRIO (1999): *Práctias de matemáticas de Bachillerato con Derive para Windows*, Ra-Ma, Madrid.
- KUTZLER, B. (1996): *Introducción a Derive para Windows*, Derisoft, Valencia. [Manual de Derive, traducción de LLORENS FUSTER, J. L. (1998).]
- PÉREZ, C. y C. PAULOGORRÁN (1998): *Matemática práctica con Derive para Windows*, Ra-Ma, Madrid.

José de Francisco Estaire
IES Andrés Laguna
Segovia

